
D02 – Ordinary Differential Equations

D02PCF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

D02PCF solves the initial value problem for a first order system of ordinary differential equations using
Runge–Kutta methods.

2 Specification

SUBROUTINE D02PCF(F, TWANT, TGOT, YGOT, YPGOT, YMAX, WORK, IFAIL)
INTEGER IFAIL
real TWANT, TGOT, YGOT(∗), YPGOT(∗), YMAX(∗), WORK(∗)
EXTERNAL F

3 Description

D02PCF and its associated routines (D02PVF, D02PYF, D02PZF) solve the initial value problem for a
first-order system of ordinary differential equations. The routines, based on Runge–Kutta methods and
derived from RKSUITE [1], integrate

y′ = f(t, y) given y(t0) = y0

where y is the vector of n solution components and t is the independent variable.

D02PCF is designed for the usual task, namely to compute an approximate solution at a sequence of
points. You must first call D02PVF to specify the problem and how it is to be solved. Thereafter you
call D02PCF repeatedly with successive values of TWANT, the points at which you require the solution,
in the range from TSTART to TEND (as specified in D02PVF). In this manner D02PCF returns the
point at which it has computed a solution TGOT (usually TWANT), the solution there (YGOT) and
its derivative (YPGOT). If D02PCF encounters some difficulty in taking a step toward TWANT, then
it returns the point of difficulty (TGOT) and the solution and derivative computed there (YGOT and
YPGOT, respectively).

In the call to D02PVF you can specify either the first step size for D02PCF to attempt or that it compute
automatically an appropriate value. Thereafter D02PCF estimates an appropriate step size for its next
step. This value and other details of the integration can be obtained after any call to D02PCF by a call
to D02PYF. The local error is controlled at every step as specified in D02PVF. If you wish to assess the
true error, you must set ERRASS = .TRUE. in the call to D02PVF. This assessment can be obtained
after any call to D02PCF by a call to D02PZF.

For more complicated tasks, you are referred to routines D02PDF, D02PXF and D02PWF, all of which
are used by D02PCF.

4 References

[1] Brankin R W, Gladwell I and Shampine L F (1991) RKSUITE: A suite of Runge–Kutta codes for
the initial value problems for ODEs SoftReport 91–S1 Southern Methodist University

5 Parameters

1: F — SUBROUTINE, supplied by the user. External Procedure

F must evaluate the functions fi (that is the first derivatives y′
i) for given values of the arguments

t, yi.

Its specification is:

SUBROUTINE F(T, Y, YP)
real T, Y(∗), YP(∗)

1: T — real Input
On entry: the current value of the independent variable, t.

2: Y(∗) — real array Input
On entry: the current values of the dependent variables, yi for i = 1, 2, . . . , n.

3: YP(∗) — real array Output
On exit: the values of fi for i = 1, 2, . . . , n.

F must be declared as EXTERNAL in the (sub)program from which D02PCF is called. Parameters
denoted as Input must not be changed by this procedure.

2: TWANT — real Input

On entry: the next value of the independent variable, t, where a solution is desired.

Constraints: TWANT must be closer to TEND than the previous value of TGOT (or TSTART on
the first call to D02PCF); see D02PVF for a description of TSTART and TEND. TWANT must
not lie beyond TEND in the direction of integration.

3: TGOT — real Output

On exit: the value of the independent variable t at which a solution has been computed. On
successful exit with IFAIL = 0, TGOT will equal TWANT. On exit with IFAIL > 1, a solution has
still been computed at the value of TGOT but in general TGOT will not equal TWANT.

4: YGOT(∗) — real array Input/Output

Note: the dimension of the array YGOT must be at least n.

On entry: on the first call to D02PCF, YGOT need not be set. On all subsequent calls YGOT must
remain unchanged.

On exit: an approximation to the true solution at the value of TGOT. At each step of the integration
to TGOT, the local error has been controlled as specified in D02PVF. The local error has still been
controlled even when TGOT �= TWANT, that is after a return with IFAIL > 1.

5: YPGOT(∗) — real array Output

Note: the dimension of the array YPGOT must be at least n.

On exit: an approximation to the first derivative of the true solution at TGOT.

6: YMAX(∗) — real array Input/Output

Note: the dimension of the array YMAX must be at least n.

On entry: on the first call to D02PCF, YMAX need not be set. On all subsequent calls YMAX
must remain unchanged.

On exit: YMAX(i) contains the largest value of |yi| computed at any step in the integration so far.

7: WORK(∗) — real array Input/Output

On entry: this must be the same array as supplied to D02PVF. It must remain unchanged between
calls.

On exit: information about the integration for use on subsequent calls to D02PCF or other associated
routines.

8: IFAIL — INTEGER Input/Output

On entry: IFAIL must be set to 0, −1 or 1. Users who are unfamiliar with this parameter should
refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error or gives a warning (see Section 6).

For this routine, because the values of output parameters may be useful even if IFAIL �= 0 on exit,
users are recommended to set IFAIL to −1 before entry. It is then essential to test the value of
IFAIL on exit.

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings specified by the routine:

IFAIL = 1

On entry, an invalid input value for TWANT was detected or an invalid call to D02PCF was made,
for example without a previous call to the setup routine D02PVF. If on entry IFAIL = 0 or −1,
the precise form of the error will be detailed on the current error message unit (as defined by
X04AAF). You cannot continue integrating the problem.

IFAIL = 2

This return is possible only when METHOD = 3 has been selected in the preceding call of D02PVF.
D02PCF is being used inefficiently because the step size has been reduced drastically many times
to get answers at many values of TWANT. If you really need the solution at this many points,
you should change to METHOD = 2 because it is (much) more efficient in this situation. To
change METHOD, restart the integration from TGOT, YGOT by a call to D02PVF. If you wish
to continue with METHOD = 3, just call D02PCF again without altering any of the arguments
other than IFAIL. The monitor of this kind of inefficiency will be reset automatically so that the
integration can proceed.

IFAIL = 3

A considerable amount of work has been expended in the (primary) integration. This is measured
by counting the number of calls to the subroutine F. At least 5000 calls have been made since the
last time this counter was reset. Calls to F in a secondary integration for global error assessment
(when ERRASS = .TRUE. in the call to D02PVF) are not counted in this total. The integration
was interrupted, so TGOT is not equal to TWANT. If you wish to continue on towards TWANT,
just call D02PCF again without altering any of the arguments other than IFAIL. The counter
measuring work will be reset to zero automatically.

IFAIL = 4

It appears that this problem is stiff. The methods implemented in D02PCF can solve such problems,
but they are inefficient. You should change to another code based on methods appropriate for stiff
problems. The integration was interrupted so TGOT is not equal to TWANT. If you want to
continue on towards TWANT, just call D02PCF again without altering any of the arguments
other than IFAIL. The stiffness monitor will be reset automatically.

IFAIL = 5

It does not appear possible to achieve the accuracy specified by TOL and THRES in the call to
D02PVF with the precision available on the computer being used and with this value of METHOD.
You cannot continue integrating this problem. A larger value for METHOD, if possible, will permit
greater accuracy with this precision. To increase METHOD and/or continue with larger values of
TOL and/or THRES, restart the integration from TGOT, YGOT by a call to D02PVF.

IFAIL = 6

(This error exit can only occur if ERRASS = .TRUE. in the call to D02PVF.) The global error
assessment may not be reliable beyond the current integration point TGOT. This may occur

because either too little or too much accuracy has been requested or because f(t, y) is not smooth
enough for values of t just past TGOT and current values of the solution y. The integration cannot
be continued. This return does not mean that you cannot integrate past TGOT, rather that you
cannot do it with ERRASS = .TRUE.. However, it may also indicate problems with the primary
integration.

7 Accuracy

The accuracy of integration is determined by the parameters TOL and THRES in a prior call to D02PVF
(see the routine document for further details and advice). Note that only the local error at each step is
controlled by these parameters. The error estimates obtained are not strict bounds but are usually reliable
over one step. Over a number of steps the overall error may accumulate in various ways, depending on
the properties of the differential system.

8 Further Comments

If D02PCF returns with IFAIL = 5 and the accuracy specified by TOL and THRES is really required
then you should consider whether there is a more fundamental difficulty. For example, the solution
may contain a singularity. In such a region the solution components will usually be large in magnitude.
Successive output values of YGOT and YMAX should be monitored (or D02PDF should be used since
this takes one integration step at a time) with the aim of trapping the solution before the singularity. In
any case numerical integration cannot be continued through a singularity, and analytical treatment may
be necessary.

Performance statistics are available after any return from D02PCF by a call to D02PYF. If ERRASS =
.TRUE. in the call to D02PVF, global error assessment is available after any return from D02PCF (except
when IFAIL = 1) by a call to D02PZF.

After a failure with IFAIL = 5 or 6 the diagnostic routines D02PYF and D02PZF may be called only
once.

If D02PCF returns with IFAIL = 4 then it is advisable to change to another code more suited to the
solution of stiff problems. D02PCF will not return with IFAIL = 4 if the problem is actually stiff but it
is estimated that integration can be completed using less function evaluations than already computed.

9 Example

We solve the equation
y′′ = −y, y(0) = 0, y′(0) = 1

reposed as
y′
1 = y2 y′

2 = −y1

over the range [0, 2π] with initial conditions y1 = 0.0 and y2 = 1.0. We use relative error control
with threshold values of 1.0E−8 for each solution component and compute the solution at intervals of
length π/4 across the range. We use a low order Runge–Kutta method (METHOD = 1) with tolerances
TOL = 1.0E−3 and TOL = 1.0E−4 in turn so that we may compare the solutions. The value of π is
obtained by using X01AAF.

Note that the length of WORK is large enough for any valid combination of input arguments to D02PVF.

See also the example program for D02PZF.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* D02PCF Example Program Text
* Mark 17 Revised. NAG Copyright 1995.
* .. Parameters ..

INTEGER NOUT
PARAMETER (NOUT=6)
INTEGER NEQ, LENWRK, METHOD
PARAMETER (NEQ=2,LENWRK=32*NEQ,METHOD=1)
real ZERO, ONE, TWO
PARAMETER (ZERO=0.0e0,ONE=1.0e0,TWO=2.0e0)

* .. Local Scalars ..
real HNEXT, HSTART, PI, TEND, TGOT, TINC, TOL, TSTART,

+ TWANT, WASTE
INTEGER I, IFAIL, J, L, NPTS, STPCST, STPSOK, TOTF
LOGICAL ERRASS

* .. Local Arrays ..
real THRES(NEQ), WORK(LENWRK), YGOT(NEQ), YMAX(NEQ),

+ YPGOT(NEQ), YSTART(NEQ)
* .. External Functions ..

real X01AAF
EXTERNAL X01AAF

* .. External Subroutines ..
EXTERNAL D02PCF, D02PVF, D02PYF, F

* .. Executable Statements ..
WRITE (NOUT,*) ’D02PCF Example Program Results’

*
* Set initial conditions and input for D02PVF
*

PI = X01AAF(ZERO)
TSTART = ZERO
YSTART(1) = ZERO
YSTART(2) = ONE
TEND = TWO*PI
DO 20 L = 1, NEQ

THRES(L) = 1.0e-8
20 CONTINUE

ERRASS = .FALSE.
HSTART = ZERO

*
* Set control for output
*

NPTS = 8
TINC = (TEND-TSTART)/NPTS

*
DO 60 I = 1, 2

IF (I.EQ.1) TOL = 1.0e-3
IF (I.EQ.2) TOL = 1.0e-4
IFAIL = 0
CALL D02PVF(NEQ,TSTART,YSTART,TEND,TOL,THRES,METHOD,

+ ’Usual Task’,ERRASS,HSTART,WORK,LENWRK,IFAIL)
*

WRITE (NOUT,’(/A,D8.1)’) ’ Calculation with TOL = ’, TOL
WRITE (NOUT,’(/A/)’) ’ t y1 y2’
WRITE (NOUT,’(1X,F6.3,2(3X,F7.3))’) TSTART, (YSTART(L),L=1,NEQ)
DO 40 J = NPTS - 1, 0, -1

TWANT = TEND - J*TINC
IFAIL = 1
CALL D02PCF(F,TWANT,TGOT,YGOT,YPGOT,YMAX,WORK,IFAIL)

*
WRITE (NOUT,’(1X,F6.3,2(3X,F7.3))’) TGOT, (YGOT(L),L=1,NEQ)

40 CONTINUE
*

IFAIL = 0
CALL D02PYF(TOTF,STPCST,WASTE,STPSOK,HNEXT,IFAIL)
WRITE (NOUT,’(/A,I6)’)

+ ’ Cost of the integration in evaluations of F is’, TOTF
*

60 CONTINUE
*

STOP
END
SUBROUTINE F(T,Y,YP)

* .. Scalar Arguments ..
real T

* .. Array Arguments ..
real Y(*), YP(*)

* .. Executable Statements ..
YP(1) = Y(2)
YP(2) = -Y(1)
RETURN
END

9.2 Program Data

None.

9.3 Program Results

D02PCF Example Program Results

Calculation with TOL = 0.1E-02

t y1 y2

0.000 0.000 1.000
0.785 0.707 0.707
1.571 0.999 0.000
2.356 0.706 -0.706
3.142 0.000 -0.999
3.927 -0.706 -0.706
4.712 -0.998 0.000
5.498 -0.705 0.706
6.283 0.001 0.997

Cost of the integration in evaluations of F is 124

Calculation with TOL = 0.1E-03

t y1 y2

0.000 0.000 1.000
0.785 0.707 0.707
1.571 1.000 0.000

2.356 0.707 -0.707
3.142 0.000 -1.000
3.927 -0.707 -0.707
4.712 -1.000 0.000
5.498 -0.707 0.707
6.283 0.000 1.000

Cost of the integration in evaluations of F is 235

